

# Qucs

---

## Test Report

SPICE to Qucs conversion: Test File 3

Mike Brinson

---

Copyright © 2007 Mike Brinson <mbrin72043@yahoo.co.uk>

Permission is granted to copy, distribute and/or modify this document under the terms of the GNU Free Documentation License, Version 1.1 or any later version published by the Free Software Foundation. A copy of the license is included in the section entitled "GNU Free Documentation License".

# Introduction

## Title

SPICE 2g6 and 3f5 resistors.

## SPICE specification

Format: SPICE 2g6<sup>1</sup>:     **RX N+ N- value [ TC=TC1 [,TC2] ]**

Notes:

1. Characters [ and ] enclose optional items
2. Resistors begin with letter R.
3. X denotes name of resistor.
4. N+ and N- are the positive and negative nodes respectively.
5. Equations:

$TNOM$  = Nominal temperature; default 27°C.

$\Delta T = T - TNOM$

$R(T) = R(TNOM) + [1 + TC1 \cdot \Delta T + TC2 \cdot \Delta T^2]$

Format: SPICE 3f5<sup>2</sup>:

1. Standard resistors: **RX N+ N- value**
2. Semiconductor resistors:  
   **RX N+ N- [value] [mname] [L=length] [W=width] [TEMP=T]**

Notes:

1. Characters [ and ] enclose optional items
2. Resistors begin with letter R.
3. X denotes name of resistors.
4. N+ and N- are the positive and negative nodes respectively.

---

<sup>1</sup>See section 6.1, SPICE 2g6 user's guide.

<sup>2</sup>See sections 3.1.1 and 3.1.2, SPICE 3f6 user's guide.

5. mname; if specified the resistance is calculated from the process information given in entry .model mname.
6. L is the length of the resistor.
7. W is the width of the resistor.
8. mname .model type R parameters:
  - TC1 : First order temperature coefficient; default  $0.0.\Omega/^{\circ}C$ .
  - TC2 : Second order temperature coefficient; default  $0.0.\Omega/^{\circ}C^2$ .
  - RSH : Sheet resistance; default -.  $\Omega$  per square.
  - DEFW : Default width; default 1e-6m.
  - NARROW : Narrowing due to side etching; default 0.0m.
  - TNOM : Nominal temperature; default 27  $^{\circ}C$ .
9. Equations  $R = RSH \cdot \frac{L - NARROW}{W - NARROW}$   
 $R(T) = R(TNOM) + [1 + TC1 \cdot \Delta T + TC2 \cdot \Delta T^2]$   
 Where  $\Delta T = T - T_0$ : T is the circuit temperature and  $T_0$  the nominal temperature.

## Test code and schematic

SPICE code: File S2Q\_test3\_a.cir

```
* SPICE to Qucs syntax test file
* SPICE 2g6 resistors.
*
.subckt S2Q_test3_a p01 p02 p03
v1 1 0 DC 1v
r1 1 p01 10k
r2 p01 0 10k
*
v2 2 0 dc 1v
r3 2 p02 10k tc=0.01
r4 p02 0 10k
*
v3 3 0 Dc 1v
```

```
r5 3 p03 10k tc=0.01 0.015
r6 p03 0 10k
.ends
.end
```




Figure 1: March 13: SPICE to Qucs conversion: Test3 schematic plus output table for SPICE 2g6 test1: linear resistors

SPICE code: File S2Q\_test3\_b.cir

```
* SPICE to Qucs syntax test file
* SPICE 3f5 resistors.
*
.subckt S2Q_test3_a p01 p02 p03
v1 1 0 DC 1v
r1 1 p01 10k
r2 p01 0 10k
*
v2 2 0 dc 1v
r3 2 p02 RMOD1 L=10u W=1u
r4 p02 0 10k
.model RMOD1 R(RSH=50 DEFW=2e-6 NARROW=1e-7)
*
.ends
.end
```

# History of simulation results

## March 13 2007, Simulation tests by Mike Brinson

A: SPICE 2g6 tests:

**Test 1** [v2, r3, v3, and r5 commented]: Linear resistors: Vp1: PASS correct output voltage.

**Test 2** [v3, and r5 commented]: SPICE 2g6 resistor with first order temperature coefficient: **FAIL** - incorrect Qucs netlist.

Qucs netlist line 14: checker error, extraneous property 'TC' is invalid in 'R:R3'

Qucs netlist:

```
# Qucs 0.0.11 /media/hda2/S2Q_test3_prj/S2Q(test3_a).sch

.Def:S2Q_test3_a _net0 _net1 _net2
Sub:X1 _net0 _net1 _net2 gnd Type="S2Q_test3_a_cir"
.Def:End

.Def:S2Q_test3_a_cir _netP01 _netP02 _netP03 _ref
.Def:S2Q_TEST3_A _ref _netP01 _netP02 _netP03
Vdc:V1 _net1 _ref U="1V"
R:R1 _net1 _netP01 R="10k"
R:R2 _netP01 _ref R="10k"
Vdc:V2 _net2 _ref U="1V"
R:R3 _net2 _netP02 R="10k" TC="0.01"
R:R4 _netP02 _ref R="10k"
R:R6 _netP03 _ref R="10k"
.Def:End
Sub:X1 _ref _netP01 _netP02 _netP03 Type="S2Q_TEST3_A"
.Def:End

Sub:SUB1 vp1 vp2 vp3 Type="S2Q_test3_a"
.DC:DC1 Temp="26.85" reltol="0.001" abstol="1_uA" vntol="1_uV"
saveOPs="no" MaxIter="150" saveAll="no" convHelper="none" Solver="CroutLU"
```

**Test 3** [v2, r2, and r3 commented]: **FAIL** - netlist not passed correctly.

Qucs error message: line 14: syntax error, unexpected Floats, expecting Eol.

Qucs netlist:

```
# Qucs 0.0.11 /media/hda2/S2Q_test3_prj/S2Q(test3_a).sch
```

```
.Def:S2Q_test3_a _net0 _net1 _net2
Sub:X1 _net0 _net1 _net2 gnd Type="S2Q_test3_a_cir"
.Def:End
```

```
.Def:S2Q_test3_a_cir _netP01 _netP02 _netP03 _ref
.Def:End
```

```
Sub:SUB1 vp1 vp2 vp3 Type="S2Q_test3_a"
.DC:DC1 Temp="26.85" reltol="0.001" abstol="1_pA"
vntol="1_uV" saveOPs="no" MaxIter="150" saveAll="no"
convHelper="none" Solver="CroutLU"
```

B: SPICE 3f5 tests

**Test1:** SPICE file `S2Q_test3_b.cir` fails to convert to Qucs netlist format, giving the following error message:  
line10 : syntax error, unexpected identifier, expecting Digits or Floats.  
SPICE 3f5 Semiconductor resistors not implemented?

## March 15 2007, Simulation tests by Mike Brinson

Qucs CVS code modifications:

- \* `scan_spice.l`: Lexer modifications for the Spice 2g6 resistor syntax were necessary. Stefan Jahn
- \* `parse_spice.y`: Allow Spice 2g6 syntax for resistors, also fixed netlist grammar for Spice 3f5 models. Stefan Jahn
- \* `check_spice.cpp`: Handle R semiconductor model correctly as well as the Spice 2g6 syntax for the temperature coefficients. Stefan Jahn

SPICE code: File `S2Q_test3_a.cir`

- Vp01.V: **PASS**; correct dc output.
- Vp02.V: **PASS**; correct dc output for TEMP=TNOM.
- Vp03.V: **PASS**; correct dc output for TEMP=TNOM.

NOTES:

- The Vp02 and Vp03 test results are only correct for TEMP=TNOM.

- SPICE 2g6 simulates circuit performance at temperature set by the value of TNOM; 27 °C by default.
- Circuits can be simulated at other temperatures by using a .TEMP control statement; which has the format

```
.TEMP T1 [ T2 [ T3 ..... ] ]
```

Unfortunately the Qucsconv program does not recognise this statement so there appears to be no way of changing the temperature of SPICE 2g6 resistors that have TC1 and TC2 temperature coefficients in their netlist entries.

- Adding SPICE 2g6 statement .TEMP 50 to file S2Q\_test3\_a.cir gives the following error:

```
spice notice, no .END directive found, continuing line 18: syntax error,
unexpected Identifier, expecting $end
```

Qucs netlist:

```
# Qucs 0.0.11 /media/hda2/S2Q_test3_prj/S2Q(test3_a).sch

.Def:S2Q_test3_a _net0 _net1 _net2
Sub:X1 _net0 _net1 _net2 gnd Type="S2Q_test3_a_cir"
.Def:End

.Def:S2Q_test3_a_cir _netP01 _netP02 _netP03 _ref
.Def:S2Q_TEST3_A _ref _netP01 _netP02 _netP03
Vdc:V1 _net1 _ref U="1V"
R:R1 _net1 _netP01 R="10k"
R:R2 _netP01 _ref R="10k"
Vdc:V2 _net2 _ref U="1V"
R:R3 _net2 _netP02 R="10k" Tc1="0.01"
R:R4 _netP02 _ref R="10k"
Vdc:V3 _net3 _ref U="1V"
R:R5 _net3 _netP03 R="10k" Tc1="0.01" Tc2="0.015"
R:R6 _netP03 _ref R="10k"
.Def:End
Sub:X1 _ref _netP01 _netP02 _netP03 Type="S2Q_TEST3_A"
.Def:End

.DC:DC1 Temp="50" reltol="0.001" abstol="1_pA" vntol="1_uV"
saveOPs="no" MaxIter="150" saveAll="no" convHelper="none"
Solver="CroutLU"
```






Figure 2: March 15: SPICE to Qucs conversion: SPICE 2g6 resistor schematic plus output table

Sub:SUB1 vp1 vp2 vp3 Type="S2Q\_test3\_a"

SPICE code: File S2Q\_test3\_b.cir

- Vp01.V: **PASS**; correct dc output.
- Vp02.V: **PASS**; correct dc output.

Qucs netlist:

```
# Qucs 0.0.11 /media/hda2/S2Q_test3_prj/S2Q(test3_b).sch

.Def:S2Q_test3_b _net0 _net1
Sub:X1 _net0 _net1 gnd Type="S2Q_test3_b_cir"
.Def:End

.Def:S2Q_test3_b_cir _netP01 _netP02 _ref
.Def:S2Q_TEST3_A _ref _netP01 _netP02
Vdc:V1 _net1 _ref U="1V"
R:R1 _net1 _netP01 R="10k"
R:R2 _netP01 _ref R="10k"
Vdc:V2 _net2 _ref U="1V"
R:R3 _net2 _netP02 R="550"
R:R4 _netP02 _ref R="10k"
.Def:End
Sub:X1 _ref _netP01 _netP02 Type="S2Q_TEST3_A"
.Def:End

Sub:SUB1 vp01 vp02 Type="S2Q_test3_b"
.DC:DC1 Temp="26.85" reltol="0.001" abstol="1_uA" vntol="1_uV"
saveOPs="no" MaxIter="150" saveAll="no" convHelper="none" Solver="CroutLU"
```

SPICE code: File S2Q\_test3\_c.cir

```
* SPICE to Qucs syntax test file
* SPICE 3f5 resistors : Temperature effects.
*
.subckt S2Q_test3_a p01 p02 p03 p04 p05 p06 p07 p08 p09 p10
v1 1 0 DC 1v
r1 1 p01 10k
r2 p01 0 10k
*
v2 2 0 dc 1v
r3 2 p02 10k RMOD1
r4 p02 0 10k
.model RMOD1 R(TC1=0.01 TC2=0.015)
```

```

*
v3 3 0 dc 1v
r5 3 p03 10k RMOD1 TEMP=30
r6 p03 0 10k
*
v4 4 0 dc 1v
r7 4 p04 10k RMOD1 TEMP=40
r8 p04 0 10k
*
v5 5 0 dc 1v
r9 5 p05 10k RMOD1 TEMP=50
r10 p05 0 10k
*
v6 6 0 dc 1v
r11 6 p06 10k RMOD1 TEMP=60
r12 p06 0 10k
*
v7 7 0 dc 1v
r13 7 p07 10k RMOD1 TEMP=70
r14 p07 0 10k
*
v8 8 0 dc 1v
r15 8 p08 10k RMOD1 TEMP=80
r16 p08 0 10k
*
v9 9 0 dc 1v
r17 9 p09 10k RMOD1 TEMP=90
r18 p09 0 10k
*
v10 10 0 dc 1v
r19 10 p10 10k RMOD1 TEMP=100
r20 p10 0 10k
.ends
.end

```

- Vp01.V: **PASS**; correct dc output.
- Vp02.V: **PASS**; correct dc output.
- Vp03.V: **PASS**; correct dc output.
- Vp04.V: **PASS**; correct dc output.
- Vp05.V: **PASS**; correct dc output.

- Vp06.V: **PASS**; correct dc output.
- Vp07.V: **PASS**; correct dc output.
- Vp08.V: **PASS**; correct dc output.
- Vp09.V: **PASS**; correct dc output.
- Vp20.V: **PASS**; correct dc output.

NOTES:

- In SPICE 3f5 TEMP values attached to resistors override the global value of TEMP.
- SPICE 3f5 differs from SPICE 2g6 in that it does not allow the control statement .TEMP.




Figure 3: March 15: SPICE to Qucs conversion: Test3 schematic plus output for SPICE 3f5 test c

Qucs netlist:

```
# Qucs 0.0.11 /media/hda2/S2Q_test3_prj/s2Q(test3_c).sch

.Def:S2Q_test3_c _net0 _net1 _net2 _net3 _net4 _net5 _net6 _net7 _net8 _net9
Sub:X1 _net0 _net1 _net2 _net3 _net4 _net5 _net6 _net7 _net8 _net9
gnd Type="S2Q_test3_c_cir"
.Def:End

.Def:S2Q_test3_c_cir _netP01 _netP02 _netP03 _netP04 _netP05
_netP06 _netP07 _netP08 _netP09 _netP10 _ref
.Def:S2Q_TEST3_A _ref _netP01 _netP02 _netP03 _netP04
_netP05 _netP06 _netP07 _netP08 _netP09 _netP10
Vdc:V1 _net1 _ref U="1V"
R:R1 _net1 _netP01 R="10k"
R:R2 _netP01 _ref R="10k"
Vdc:V2 _net2 _ref U="1V"
R:R3 _net2 _netP02 R="10k" Tc1="0.01" Tc2="0.015"
R:R4 _netP02 _ref R="10k"
Vdc:V3 _net3 _ref U="1V"
R:R5 _net3 _netP03 R="10k" Temp="30" Tc1="0.01" Tc2="0.015"
R:R6 _netP03 _ref R="10k"
Vdc:V4 _net4 _ref U="1V"
R:R7 _net4 _netP04 R="10k" Temp="40" Tc1="0.01" Tc2="0.015"
R:R8 _netP04 _ref R="10k"
Vdc:V5 _net5 _ref U="1V"
R:R9 _net5 _netP05 R="10k" Temp="50" Tc1="0.01" Tc2="0.015"
```

```

R:R10 _netP05 _ref R="10k"
Vdc:V6 _net6 _ref U="1V"
R:R11 _net6 _netP06 R="10k" Temp="60" Tc1="0.01" Tc2="0.015"
R:R12 _netP06 _ref R="10k"
Vdc:V7 _net7 _ref U="1V"
R:R13 _net7 _netP07 R="10k" Temp="70" Tc1="0.01" Tc2="0.015"
R:R14 _netP07 _ref R="10k"
Vdc:V8 _net8 _ref U="1V"
R:R15 _net8 _netP08 R="10k" Temp="80" Tc1="0.01" Tc2="0.015"
R:R16 _netP08 _ref R="10k"
Vdc:V9 _net9 _ref U="1V"
R:R17 _net9 _netP09 R="10k" Temp="90" Tc1="0.01" Tc2="0.015"
R:R18 _netP09 _ref R="10k"
Vdc:V10 _net10 _ref U="1V"
R:R19 _net10 _netP10 R="10k" Temp="100" Tc1="0.01" Tc2="0.015"
R:R20 _netP10 _ref R="10k"
.Def:End
Sub:X1 _ref _netP01 _netP02 _netP03 _netP04 _netP05 _netP06 _netP07
_netP08 _netP09 _netP10 Type="S2Q_TEST3_A"
.Def:End

```

```

Sub:SUB1 vp01 vp02 vp03 vp04 vp05 vp06 vp07 vp08 vp09 vp10 Type="S2Q_test3_c"
.DC:DC1 Temp="26.8" reltol="0.001" abstol="1_uA" vntol="1_uV" saveOPs="no"
MaxIter="150" saveAll="no" convHelper="none" Solver="CroutLU"

```

## March 18 2007, Simulation tests by Mike Brinson

More SPICE 3f5 Temperature tests.

SPICE code: File S2Q\_test3\_f.cir

```

* SPICE to Qucs syntax test file
* SPICE 3f5 resistors.
* Temperature tests.
*
.subckt S2Q_test3_f p01 p02 p03 p04
v1 1 0 DC 1v
r1 1 p01 10k
r2 p01 0 10k
*
*
v2 2 0 dc 1v
r3 2 p02 10k RMOD1 TEMP=50

```

```

r4 p02 0 10k
.model RMOD1 R(TC1=0.01 TC2=0.015)
*
v3 3 0 dc 1v
r5 3 p03 10k RMOD1
r6 p03 0 10k
.model RMOD1 R(TC1=0.01 TC2=0.015 TNOM=100)
*
v4 4 0 dc 1v
r7 4 p04 10k RMOD1
r8 p04 0 10k
.model RMOD1 R(TC1=0.01 TC2=0.015)
*
.ends
.OPTION TNOM=40
.end

```

- Vp01.V: **PASS**; correct dc output.
- Vp02.V: **PASS**; correct dc output.
- Vp03.V: **FAIL**; dc output does not change with changes in .OPTION TEMP or TNOM.
- Vp04.V: **FAIL**; dc output does not change with changes in .OPTION TEMP or TNOM.

NOTES:

- In SPICE 3f5 TEMP values attached to resistors override the global value of circuit temperature
- Output Vp02 is correct the value of R3 being determined by TEMP=50, TC1=0.01 and TC2=0.015 during the resistance calculation.
- Using SPICE statements .OPTION TNOM=XX or .OPTION TEMP=XX appears to have no effect on resistance calculations.
- Similarly, adding TNOM=XX to a R model appears not to affect the resistance calculations.




Figure 4: March 18: SPICE to Qucs conversion: Test3 schematic plus output for SPICE 3f5 test f

Qucs netlist:

```
# Qucs 0.0.11 /media/hda2/S2Q_test3_prj/S2Q(test3_f).sch

.Def:S2Q_test3_f _net0 _net1 _net2 _net3
Sub:X1 _net0 _net1 _net2 _net3 gnd Type="S2Q_test3_f_cir"
.Def:End

.Def:S2Q_test3_f_cir _netP01 _netP02 _netP03 _netP04 _ref
.Def:S2Q_TEST3_F _ref _netP01 _netP02 _netP03 _netP04
Vdc:V1 _net1 _ref U="1V"
R:R1 _net1 _netP01 R="10k"
R:R2 _netP01 _ref R="10k"
Vdc:V2 _net2 _ref U="1V"
R:R3 _net2 _netP02 R="10k" Temp="50" Tc1="0.01" Tc2="0.015"
R:R4 _netP02 _ref R="10k"
Vdc:V3 _net3 _ref U="1V"
R:R5 _net3 _netP03 R="10k" Tc1="0.01" Tc2="0.015"
R:R6 _netP03 _ref R="10k"
Vdc:V4 _net4 _ref U="1V"
R:R7 _net4 _netP04 R="10k" Tc1="0.01" Tc2="0.015"
R:R8 _netP04 _ref R="10k"
.Def:End
Sub:X1 _ref _netP01 _netP02 _netP03 _netP04 Type="S2Q_TEST3_F"
.Def:End

Sub:SUB1 vp01 vp02 vp03 vp04 Type="S2Q_test3_f"
```



```
.DC:DC1 Temp="26.8" reltol="0.001" abstol="1_pA" vntol="1_uV"
saveOPs="no" MaxIter="150" saveAll="no" convHelper="none" Solver="CroutLU"
```

## March 25 2007, Simulation tests by Mike Brinson

Code modifications:

- \* `scan_spice.l`, `parse_spice.y`: Accept `.TEMP` syntax (Spice 2g6) in lexer and parser. Stefan Jahn.

SPICE code: File `S2Q_test3_a.cir`

SPICE statements `.OPTION TNOM=50` and `.TEMP 50` are now accepted in the SPICE to Qucs translation process. HOWEVER, at the moment (Qucs 0.0.12) global circuit temperature parameters are not implemented in the Qucs simulator and cannot be changed via these statements.

SPICE code: File `S2Q_test3_f.cir`

SPICE code modification due to test bug caused by `RMOD1` being referenced in each test.

New code:

```
* SPICE to Qucs syntax test file
* SPICE 3f5 resistors.
* Temperature tests.
*
.subckt S2Q_test3_f p01 p02 p03 p04
v1 1 0 DC 1v
r1 1 p01 10k
r2 p01 0 10k
v2 2 0 dc 1v
r3 2 p02 10k RMOD1 TEMP=50
r4 p02 0 10k
.model RMOD1 R(TC1=0.01 TC2=0.015)
v3 3 0 dc 1v
r5 3 p03 10k RMOD2
r6 p03 0 10k
.model RMOD2 R(TC1=0.01 TC2=0.015 TNOM=100)
v4 4 0 dc 1v
r7 4 p04 10k RMOD3
r8 p04 0 10k
.model RMOD3 R(TC1=0.01 TC2=0.015)
.ends
```




Figure 5: March 25: SPICE to Qucs conversion: Test3 schematic plus output for SPICE 3f5 test f

```
.OPTION TNOM=40
.end
```

Qucs netlist:

```
# Qucs 0.0.12 /media/hda2/S2Q_test3_prj/S2Q(test3_f).sch

.Def:S2Q_test3_f _net0 _net1 _net2 _net3
Sub:X1 _net0 _net1 _net2 _net3 gnd Type="S2Q_test3_f_cir"
.Def:End

.Def:S2Q_test3_f_cir _netP01 _netP02 _netP03 _netP04 _ref
.Def:S2Q_TEST3_F _ref _netP01 _netP02 _netP03 _netP04
Vdc:V1 _net1 _ref U="1V"
R:R1 _net1 _netP01 R="10k"
R:R2 _netP01 _ref R="10k"
Vdc:V2 _net2 _ref U="1V"
R:R3 _net2 _netP02 R="10k" Temp="50" Tc1="0.01" Tc2="0.015"
R:R4 _netP02 _ref R="10k"
Vdc:V3 _net3 _ref U="1V"
R:R5 _net3 _netP03 R="10k" Tc1="0.01" Tc2="0.015" Tnom="100"
R:R6 _netP03 _ref R="10k"
Vdc:V4 _net4 _ref U="1V"
R:R7 _net4 _netP04 R="10k" Tc1="0.01" Tc2="0.015"
R:R8 _netP04 _ref R="10k"
.Def:End
Sub:X1 _ref _netP01 _netP02 _netP03 _netP04 Type="S2Q_TEST3_F"
.Def:End

Sub:SUB1 vp01 vp02 vp03 vp04 Type="S2Q_test3_f"
.DC:DC1 Temp="26.8" reltol="0.001" abstol="1_uA" vntol="1_uV"
saveOPs="no" MaxIter="150" saveAll="no" convHelper="none" Solver="CroutLU"
```

- Vp01.V: **PASS**; correct dc output.
- Vp02.V: **PASS**; correct dc output.
- Vp03.V: **PASS**; correct dc output.
- Vp04.V: **FAIL**; dc output does not change with changes in .OPTION TNOM = 40. The Qucs simulator does NOT implement a global temperature parameter but allocates separate temperatures to each component.

NOTES:

- In SPICE 3f5 TEMP values attached to resistors override the global value of circuit temperature
- Output Vp02 is correct, the value of R3 being determined by TEMP=50, TC1=0.01 and TC2=0.015.
- Output Vp03 is correct, the value of R5 being determined by TNOM=100, TC1=0.01 and TC2=0.015.
- Output VP04 is 0.5 because .OPTION TNOM=40 has no effect on the temperature of resistor R8. By default the temperature of R8 is set to 26.85°C.

## References

1. A. Vladimirescu, Kaihe Zhang, A.R. Newton, D.O Pederson A. Sangiovanni-Vincentelli, SPICE 2G User's Guide (10 Aug 1981), Department of Electrical Engineering and Computer Sciences, University of California, Berkeley, Ca., 94720.
2. B. Johnson, T. Quarles, A.R. Newton, P.O. Pederson, A.Sangiovanni-Vincentelli, SPICE3 Version 3f User's Manual (October 1972), Department of Electrical Engineering and Computer Sciences, University of California, Berkeley, Ca., 94720.
3. Andrei Vladimirescu, THE SPICE book,1994, John Wiley and Sons. Inc., ISBN 0-471-609-26-9.